2-hour test in 01006 Advanced Engineering Mathematics
May 11 2022

shsp 10.5.22

Problem 1

```
> restart: with(LinearAlgebra):
```

About a function $f: \mathbb{R}^2 \to \mathbb{R}$ it is given that the two first-order partial derivatives are

```
> fx:=6*x-6*y
```

$$fx := 6\,x - 6\,y \tag{1.1}$$

and

```
> fy:=6*y^2-6*x
```

$$fy := 6\,y^2 - 6\,x \tag{1.2}$$

Question 1

$\nabla f(x, y) = (f'_x(x, y), f'_y(x, y)) = (6\,x - 6\,y, 6\,y^2 - 6\,x) = (0, 0) \quad 6\,x - 6\,y = 0$ and $6\,y^2 - 6\,x = 0$
$x = y$ and $y(y-1) = 0 \quad y = 0$ and $x = 0$ or $y = 1$ and $x = 1$.
All stationary points for f are then $(0, 0)$ and $(1, 1)$.

Question 2

The 2nd-order partial derivatives of f are

```
> fxx:=diff(fx,x);fxy:=diff(fx,y);fyy:=diff(fy,y);
```

$$fxx := 6$$
$$fxy := -6$$
$$fyy := 12\,y \tag{1.2.1}$$

The Hessian matrix of f at point (x, y) is

```
> H(x,y):=<fxx,fxy;fxy,fyy>
```

$$H(x, y) := \begin{bmatrix} 6 & -6 \\ -6 & 12\,y \end{bmatrix} \tag{1.2.2}$$

If f has local extremum at a point then that point must be a stationary point since f has no exceptional points.

```
> H(0,0):=subs(x=0,y=0,H(x,y))
```

$$H(0, 0) := \begin{bmatrix} 6 & -6 \\ -6 & 0 \end{bmatrix} \tag{1.2.3}$$

```
> Eigenvalues(H(0,0),output=list)
```

$$\left[3 + 3\sqrt{5}, 3 - 3\sqrt{5}\right] \tag{1.2.4}$$

As the two eigenvalues of $\mathbf{H}(0, 0)$ have opposite signs, we can conclude that f has neither local maximum nor local minimum in the stationary point $(0, 0)$.

```
> H(1,1):=subs(x=1,y=1,H(x,y))
```

$$H(1,1) := \begin{bmatrix} 6 & -6 \\ -6 & 12 \end{bmatrix} \quad (1.2.5)$$

```
> Eigenvalues(H(1,1),output=list)
```

$$[9+3\sqrt{5}, 9-3\sqrt{5}] \quad (1.2.6)$$

As both eigenvalues of $\mathbf{H}(1, 1)$ are positive, we can conclude that f has a proper local minimum in the stationary point $(1, 1)$.

Question 3

It is now given that $f(0, 0) = 1$.
The approximating 2nd-degree polynomial of f with expansion point (0, 0) is
$P_2(x, y) =$

$f(0,0) + f'_x(0,0)x + f'_y(0,0)y + \frac{1}{2}f''_{xx}(0,0)x^2 + f''_{xy}(0,0)xy + \frac{1}{2}f''_{yy}(0,0)y^2 =$

$1 + 3x^2 - 6xy$.

Problem 2

```
> restart:
  with(plots):
  prik:=(x,y)->VectorCalculus[DotProduct](x,y):
  kryds:=(x,y)->convert(VectorCalculus[CrossProduct](x,y),Vector):
  `vekdif`:=proc(X,Y) convert(diff(convert(X,list),Y),Vector) end
  proc:
  `vop`:=proc(X) op(convert(X,list)) end proc:
```

A function $h : \mathbb{R}^2 \to \mathbb{R}$ is given by the expression

```
> h:=(x,y)->2*x-y+1
```

$$h := (x, y) \mapsto 2\cdot x - y + 1 \quad (2.1)$$

The rectangle M_1 is located in the (x, y) place in (x, y, z) space and is given by
$M_1 = \{(x, y) \mid 0 \le x \le 2, 0 \le y \le 1\}$.
Let G_1 denote the part of the graph of h that is located vertically above M_1.

Question 1

A parametric representation of G_1 is

```
> r:= (u,v)-><u,v,2*u-v+1>:
> r(u,v)
```

$$\begin{bmatrix} u \\ v \\ 2u - v + 1 \end{bmatrix} \quad (2.1.1)$$

where $u \in [0; 2]$ and $v \in [0; 1]$.
The normal vector of the surface is

```
> N:=kryds(diff(r(u,v),u),diff(r(u,v),v))
```

$$N := \begin{bmatrix} -2 \\ 1 \\ 1 \end{bmatrix} \tag{2.1.2}$$

The corresponding Jacobian function is

```
> Jacobian:=sqrt(prik(N,N))
```

$$Jacobian := \sqrt{6} \tag{2.1.3}$$

$$\mathrm{Ar}(G_1) = \int_{G_1} 1 \, \mathrm{d}\mu = \int_0^1 \int_0^2 \mathrm{Jacobian}(u, v) \mathrm{d}u \, \mathrm{d}v = \int_0^1 \int_0^2 \sqrt{6} \, \mathrm{d}u \, \mathrm{d}v = 2\sqrt{6} \, .$$

The straight line segment between the points (0, 1) and (2, 0) divide M_1 in to parts. Let M_2 denote the lower part and let G_2 denote the part of the graph of h that is located vertically above M_2 .

Question 2

The straight line segment between the points (0, 1) and (2, 0) has the equation $y = 1 - \frac{x}{2}$.

A parametric representation of M_2 is then

$\mathbf{r}(u, v) = (u, 0) + v(0, 1 - \frac{u}{2}) = (u, v(1 - \frac{u}{2}))$, where $u \in [0\,;2]$ and $v \in [0\,;1]$.

From this we construct the following parametric representation of G_2

```
> r:=(u,v)-><u,v*(1-u/2),2*u-v*(1-u/2)+1>:
> r(u,v)
```

$$\begin{bmatrix} u \\ v\left(1 - \frac{u}{2}\right) \\ 2u - v\left(1 - \frac{u}{2}\right) + 1 \end{bmatrix} \tag{2.2.1}$$

where $u \in [0; 2]$ and $v \in [0\,;\,1]$.

The normal vector of the surface is

```
> N:=simplify(kryds(diff(r(u,v),u),diff(r(u,v),v)))
```

$$N := \begin{bmatrix} -2 + u \\ 1 - \frac{u}{2} \\ 1 - \frac{u}{2} \end{bmatrix} \tag{2.2.2}$$

The corresponding Jacobian function is

```
> Jacobian:=sqrt(prik(N,N)) assuming -2+u<0
```

$$Jacobian := \frac{\sqrt{6}\,(2 - u)}{2} \tag{2.2.3}$$

Question 3

Let f be given by

```
> f:=(x,y,z)->x+y+z-1:
> f(x,y,z)
```

$$x + y + z - 1 \tag{2.3.1}$$

here $(x, y, z) \in \mathbb{R}^3$.

The wanted surface integral is

$$\int_{G_2} f \, d\mu = \int_0^1 \int_0^2 f(\mathbf{r}(u, v)) \, \text{Jacobi}(u, v) \, du \, dv$$

```
> integrand:=f(vop(r(u,v)))*Jacobian
```

$$integrand := \frac{3 u \sqrt{6} (2 - u)}{2} \tag{2.3.2}$$

```
> Int(Int(integrand,u=0..2),v=0..1)=int(int(integrand,u=0..2),v=0..1)
```

$$\int_0^1 \int_0^2 \frac{3 u \sqrt{6} (2 - u)}{2} \, du \, dv = 2 \sqrt{6} \tag{2.3.3}$$

Problem 3

```
> restart:with(LinearAlgebra):with(plots):
```

A solid body L in (x, y, z) space is given by the parametric representation

```
> r:=(u,v,w)-><v*u^2*cos(w),v*u^2*sin(w),u>:
> r(u,v,w)
```

$$\begin{bmatrix} v u^2 \cos(w) \\ v u^2 \sin(w) \\ u \end{bmatrix} \tag{3.1}$$

where $u \in [0; 1]$, $v \in [0; 1]$ and $w \in [0; \frac{\pi}{2}]$ (see the figure in the question text).

Question 1

```
> M:=<diff(r(u,v,w),u)|diff(r(u,v,w),v)|diff(r(u,v,w),w)>
```

$$M := \begin{bmatrix} 2 v u \cos(w) & u^2 \cos(w) & -v u^2 \sin(w) \\ 2 v u \sin(w) & u^2 \sin(w) & v u^2 \cos(w) \\ 1 & 0 & 0 \end{bmatrix} \tag{3.1.1}$$

```
> Jr:=simplify(Determinant(M))
```

$$Jr := u^4 v \tag{3.1.2}$$

which is greater than or equal to zero since $v \geq 0$. The corresponding Jacobian is thus

```
> Jacobian:=Jr
```

$$Jacobian := u^4 v \quad (3.1.3)$$

Question 2

We consider the vector field

```
> V:=(x,y,z)-><x+exp(y*z),2*y-exp(x*z),3*z+exp(x*y)>:
> V(x,y,z)
```

$$\begin{bmatrix} x + e^{yz} \\ 2y - e^{xz} \\ 3z + e^{yx} \end{bmatrix} \quad (3.2.1)$$

∂L is the closed surface of L with an orientation according to outwards-pointing unit normal vector. From the Gauss Divergence Theorem we then get

$$\mathrm{Flux}(\mathbf{V}, \partial L) = \int_L \mathrm{Div}(\mathbf{V})\, \mathrm{d}\mu = \int_0^{\frac{\pi}{2}} \int_0^1 \int_0^1 \mathrm{Div}(\mathbf{V})(\mathbf{r}(u, v, w))\mathrm{Jacobian}(u, v, w)\, \mathrm{d}u\, \mathrm{d}v\, \mathrm{d}w$$

```
> div:=V->VectorCalculus[Divergence](V):
> DivV:=div(V)(x,y,z)
```

$$DivV := 6 \quad (3.2.2)$$

```
> integrand:=DivV*Jacobian
```

$$integrand := 6 u^4 v \quad (3.2.3)$$

```
> Int(integrand,[u=0..1,v=0..1,w=0..Pi/2])=int(integrand,[u=0..1,v=
  0..1,w=0..Pi/2])
```

$$\int_0^{\frac{\pi}{2}} \int_0^1 \int_0^1 6 u^4 v \, \mathrm{d}u\, \mathrm{d}v\, \mathrm{d}w = \frac{3\pi}{10} \quad (3.2.4)$$

Question 3

A parametric representation of a profile area F in the (x, z) plane which forms the solid body L according to the description in the question text is achieved by setting $w = 0$ in th egiven parametric representation of L.

```
> s:=(u,v)-><v*u^2,0,u>:
> s(u,v)
```

$$\begin{bmatrix} v u^2 \\ 0 \\ u \end{bmatrix} \quad (3.3.1)$$

where $u \in [0; 1]$ and $v \in [0; 1]$.

```
> plot3d(s(u,v),u=0..1,v=0..1,scaling=constrained,axes=normal,view=
  0..1,orientation=[-120,70],color=yellow)
```



Problem 4

```
> restart:with(plots):with(LinearAlgebra):
  prik:=(x,y)->VectorCalculus[DotProduct](x,y):
  vop:=proc(X)op(convert(X,list))end proc:
```

Let a be a positive real number. A flowcurve K of a smooth vector field **V** in (x, y, z) space is given by the parametric representation

```
> r:=t-><exp(-t),cos(t)-sin(t),cos(t)+sin(t)>:
> r(t)
```

$$\begin{bmatrix} e^{-t} \\ \cos(t) - \sin(t) \\ \cos(t) + \sin(t) \end{bmatrix} \quad \textbf{(4.1)}$$

where $t \in [0; a]$.

Question 1

The corresponding Jacobian is

```
> Jacobian:=simplify(sqrt(prik(diff(r(t),t),diff(r(t),t))))
```

$$Jacobian := \sqrt{e^{-2t} + 2} \quad \textbf{(4.1.1)}$$

which means that $m = 2$ and $n = 2$.

Question 2

Since K is a flow curve for the vector field **V**, then we have the relationship $\mathbf{V}(\mathbf{r}(t)) = \mathbf{r}'(t)$ for all $t \in [0;$ a]. From this it follows that

$$\mathrm{Tan}(\mathbf{V}, K) = \int_K \mathbf{V} \cdot \mathbf{e}\, d\mu = \int_0^a \mathbf{V}(\mathbf{r}(t)) \cdot \mathbf{r}'(t)\, dt = \int_0^a \mathbf{r}'(t) \cdot \mathbf{r}'(t) dt = \int_0^a (\mathrm{Jacobian}(t))^2\ dt\ .$$

Question 3

Let $a = 5$, and then we can calculate

```
> Tan:=Int(Jacobian^2,t=0..5)=int(Jacobian^2,t=0..5)
```

$$Tan := \int_0^5 \left(e^{-2t} + 2\right) dt = \frac{21}{2} - \frac{e^{-10}}{2} \qquad \textbf{(4.3.1)}$$

```
> curve:=spacecurve(r(t),t=0..5,color=red,axes=normal):
  pnt:=pointplot3d([[1,1,1],[vop(r(5))]],symbol=solidcircle,
  symbolsize=15):
> display(curve,pnt,tickmarks=[3,3,3],scaling=constrained,
  orientation=[-65,80,0]);
```




```
>
```